University of Warwick, Department of Sociology, 2014/15
SO201: SSAASS Surveys and Statistics (Richard Lampard)

Handout for Week 7: A D-I-Y Practice (Multiple) Linear Regression Exercise
This exercise focuses on the basics of (multiple) regression, and what it shows us in multivariate analysis terms, rather than paying much attentions to assumptions… You can tell whether you have used the menu for a command ‘correctly’ by pasting it to a syntax window (click on ‘Paste’) and seeing whether it matches the syntax under each menu choice below. First of all, you need to download the GHS91 example data, which can be found at: http://go.warwick.ac.uk/so201/data).
Recode the working hours variable (“Hours usually worked per week”) to remove a value that is not ‘real’:

** ANALYZE / DESCRIPTIVE STATISTICS / FREQUENCIES

FREQUENCIES VARIABLES=workhrs

  /ORDER=ANALYSIS.

** TRANSFORM / RECODE INTO SAME VARIABLES

RECODE workhrs (99=SYSMIS).

EXECUTE.

Then produce the correlation coefficient for usual working hours (WORKHRS) and usual gross weekly earnings (UGE):

** ANALYZE / CORRELATE / BIVARIATE

CORRELATIONS

  /VARIABLES=uge workhrs

  /PRINT=TWOTAIL NOSIG

  /MISSING=PAIRWISE.

The correlation coefficient is positive and statistically significant (p<0.05), so one can conclude that the two variables are positively correlated in the population. However, we need to do a regression to find out the amount of extra earnings that are generated by an extra hour of work:

** ANALYZE / REGRESSION / LINEAR

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT uge

  /METHOD=ENTER workhrs.

The SPSS output for a linear regression is extensive, and not all immediately relevant.
· R, in the case of a bivariate regression analysis (i.e. a regression analysis with one explanatory variable), is equivalent to the correlation coefficient.
· R Square is r-squared, and can be viewed as the proportion of the variation in the dependent variable explained by the explanatory (independent) variable(s).

· The Analysis of Variance table, for a regression analysis with one explanatory variable, is basically an F-test of significance for the correlation coefficient.

· The value of B next to “Hours usually worked per week” is the (estimated) increase in the dependent variable corresponding to a unit increase in an explanatory (independent) variable. Std. Error is the standard deviation of the estimated increase, i.e. a measure of how accurate it is. B/(Std. Error) gives a t-statistic (t), which, if greater than (approximately) 2, or less than (approximately) -2, indicates that the effect (B) is statistically significant (i.e. significantly different from zero; this can be judged more precisely by whether the value of Sig {the p-value} is less than 0.05). 


In the case of a regression analysis with one explanatory variable, the square of the value of t is equal to the F-statistic in the Analysis of Variance table. This is because testing the correlation coefficient for significance and testing the effect of one variable on the other variable for significance is effectively the same test, i.e. testing whether the two variables are related to each other.

· The value of B next to (Constant) is the predicted value for the earnings of someone who works zero hours per week! The corresponding t-statistic and its Sig. value (p-value) can be used to assess whether this differs significantly from zero.

We can use categorical variables as explanatory variables in a regression equation as long as they only have two categories, since a variable with two categories is implicitly a interval-level scale! If we want to look at the effect of gender on earnings we need to recode the variable SEX so that one value is 1 and the other is 0 (zero).

** TRANSFORM / RECODE INTO DIFFERENT VARIABLES

RECODE sex (1=1) (2=0) INTO sex2.

EXECUTE.

In a multiple regression more than one independent (explanatory) variable is used to explain variation in the dependent variable. In this case we now look simultaneously at the effects of WORKHRS and SEX2 on our dependent variable, UGE.

** ANALYZE / REGRESSION / LINEAR

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT uge

  /METHOD=ENTER workhrs sex2.

Since the t-value (t) for sex (SEX2) is greater than 2, and has a Sig. (p) value of less than 0.05, the gender effect is significant. Note also that the R Square has risen, i.e. the proportion of the variation in earnings which has been explained has risen. Furthermore, the value of B for “Hours usually worked per week” has declined, indicating some of the apparent effect of working hours in the previous equation is better explained by gender.

We may now wish to include age (AGE) in our multiple regression:

** ANALYZE / REGRESSION / LINEAR

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT uge

  /METHOD=ENTER workhrs sex2 age.

Though age can be seen to have a significant positive effect, we can see from a scatterplot that the relationship between earnings and age is non-linear. However, a curved relationship can be represented by a combination of age and the square of age.

** GRAPHS / LEGACY DIALOGS / SCATTER/DOT (Simple & Define)
GRAPH

  /SCATTERPLOT(BIVAR)=age WITH uge

  /MISSING=LISTWISE.

** TRANSFORM / COMPUTE VARIABLE
COMPUTE age2=age * age.

EXECUTE.

** ANALYZE / REGRESSION / LINEAR

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT uge

  /METHOD=ENTER workhrs sex2 age age2.

Both the effects of age and squared age (AGE2) are significant, but these two explanatory variables need to be considered together. In combination they model the age/earnings relationship in terms of a curve which is upwards sloping until age 40, relatively flat until 55, and then downwards sloping from then on, i.e. the actual pattern in the scatterplot. 

We now might want to take account of (occupational) social class. First of all we need to create a dichotomous (two-category) version, e.g. by creating a non-manual category and a manual category.

** TRANSFORM / RECODE INTO DIFFERENT VARIABLES

RECODE soclase (1 thru 3=1) (4 thru 6=0) (ELSE=SYSMIS) INTO sc2.

EXECUTE.

** ANALYZE / REGRESSION / LINEAR

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT uge

  /METHOD=ENTER workhrs sex2 age age2 sc2.

Note that the effect of sex (i.e. the value of B for SEX2) has changed quite markedly from what it was in the previous regression, and that the effect continues to fluctuate in the later regressions. (This is because gender effects can both be explained in terms of and also be concealed by the effects of other variables).

Shifting our attention back to class effects, however, the earnings of people in the highest class may be particularly high, so we may wish to set up another dichotomous variable, contrasting Class I with all other classes. (Such dichotomous variables with values of 0 and 1 are often referred to as ‘dummy variables’).

** TRANSFORM / RECODE INTO DIFFERENT VARIABLES

RECODE soclase (1=1) (2 thru 6=0) (ELSE=SYSMIS) INTO sc3.

EXECUTE.

** ANALYZE / REGRESSION / LINEAR

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT uge

  /METHOD=ENTER workhrs sex2 age age2 sc2 sc3.

Note that the value of B relating to SC3 is in effect the difference between Class I people and Class II people. The difference between Class I people and Class IIIM to V people is the sum of the values of B corresponding to SC2 and SC3.

We may want to take account of highest educational qualifications. One possibility is to contrast ‘A’ levels and above with all lower educational levels:

** ANALYZE / DESCRIPTIVE STATISTICS / FREQUENCIES

FREQUENCIES VARIABLES=edlev

  /ORDER=ANALYSIS.

** TRANSFORM / RECODE INTO DIFFERENT VARIABLES

RECODE edlev (1 thru 7=1) (8 thru 17=0) (ELSE=SYSMIS) INTO ed2.

EXECUTE.

** ANALYZE / REGRESSION / LINEAR

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT uge

  /METHOD=ENTER workhrs sex2 age age2 sc2 sc3 ed2.

Note that the values corresponding to SC2 and SC3 have diminished, indicating that some of what appeared to be an occupational class effect is, at root, an educational effect. (Note also that all the above explanatory (independent) variables together have only explained just over 0.22 (22%) of the variation in earnings. So we’ve only told a small part of the story... although we could create dummy variables for all the class/education categories!) 

